private ["_marker","_make","_trigger","_activation"];
params ["_markerName","_markerPosition","_markerType",["_color","ColorRed"],["_hidden",false]];

_marker = createMarker [_markerName, _markerPosition];
_marker setMarkerShape "ICON";
_marker setMarkerColor _color;

if (isNil "Escape_intel_find") then {
	_make = createMarker ["intel" + _markerName, _markerPosition]; //
	_make setMarkerShape "ELLIPSE";
	_make setMarkerSize [0,0];
	_make setMarkerColor "ColorRed";
	_make setMarkerBrush "Border";
	Escape_intelMaks pushback [_markerPosition,_make];
};

if (Param_RevealMarkers == 0) then {_marker setMarkerType _markerType; _make setMarkerSize [Escape_intel_Range, Escape_intel_Range]};
if (Param_RevealMarkers == 1 && {!_hidden}) then {
	_marker setMarkerType "hd_unknown";
	_make setMarkerSize [Escape_intel_Range, Escape_intel_Range];
	_trigger = createTrigger["EmptyDetector", _markerPosition, false];
	_trigger setTriggerArea[200, 200, 0, false];
	_trigger setTriggerActivation[A3E_VAR_Side_Blufor_Str, "PRESENT", false];
	_activation = format["%1 setmarkerType %2;",str _marker,str _markerType];
	_trigger setTriggerStatements["this && {isServer}",_activation ,""];
};
if(Param_RevealMarkers == 2 || {_hidden}) then {
	_marker setMarkerType "Empty";
	_trigger = createTrigger["EmptyDetector", _markerPosition, false];
	_trigger setTriggerArea[200, 200, 0, false];
	_trigger setTriggerActivation["GROUP", "PRESENT", false];
	_trigger triggerAttachVehicle (call A3E_fnc_GetPlayers - Escape_Players_enemy);
	_activation = format["%1 setmarkerType %2; %3 setMarkerSize [%4,%4];", str _marker,str _markerType,str _make,Escape_intel_Range];
	_trigger setTriggerStatements["this && {isserver}",_activation ,""];
};
if (Param_RevealMarkers == 3) then {deleteMarker _marker; deleteMarker _make; Escape_intelMaks = Escape_intelMaks - [[_markerPosition,_make]]};

if (Param_RevealMarkers > 3) then {diag_log "Unknown type of Param_RevealMarkers"};


